Problem:
Score : 500 points

Problem Statement

There are N towns on a plane. The i-th town is located at the coordinates (x_i,y_i). There may be more than one town at the same coordinates.

You can build a road between two towns at coordinates (a,b) and (c,d) for a cost of min(|a-c|,|b-d|) yen (the currency of Japan). It is not possible to build other types of roads.

Your objective is to build roads so that it will be possible to travel between every pair of towns by traversing roads. At least how much money is necessary to achieve this?

Constraints

2 ≤ N ≤ 10^5

0 ≤ x_i,y_i ≤ 10^9

All input values are integers.

Input

Input is given from Standard Input in the following format:

N
x_1 y_1
x_2 y_2
:
x_N y_N

Output

Print the minimum necessary amount of money in order to build roads so that it will be possible to travel between every pair of towns by traversing roads.

Sample Input 1

3
1 5
3 9
7 8

Sample Output 1

3

Build a road between Towns 1 and 2, and another between Towns 2 and 3. The total cost is 2+1=3 yen.

Sample Input 2

6
8 3
4 9
12 19
18 1
13 5
7 6

Sample Output 2

8

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import Data.Array.IO
import Data.List
import Data.Ord
import qualified Data.ByteString.Char8 as BS
import Data.Maybe
import Control.Monad

type UnionFindIO = IOArray Int (Int,Int)

newUFIO :: Int -> IO UnionFindIO
newUFIO n = newListArray (1,n) (zip [1..n] (repeat 1))

findUFIO :: Int -> UnionFindIO -> IO Int
findUFIO x uf = do
  (parent,size) <- readArray uf x
  if parent == x then
    return x
    else do
    next <- findUFIO parent uf
    writeArray uf x (next,size)
    return next

sameUFIO :: Int -> Int -> UnionFindIO -> IO Bool
sameUFIO x y uf = do
  xx <- findUFIO x uf
  yy <- findUFIO y uf
  return $ xx == yy

uniteUFIO :: Int -> Int -> UnionFindIO -> IO ()
uniteUFIO x y uf = do
  xx <- findUFIO x uf
  yy <- findUFIO y uf
  (_,xs) <- readArray uf xx
  (_,ys) <- readArray uf yy

  if xx == yy then
    return ()
    else if xs < ys then do
    writeArray uf xx (yy,0)
    else if xs > ys then do
    writeArray uf yy (xx,0)
    else do
    writeArray uf yy (xx,0)
    writeArray uf xx (xx,xs+1)


type City = (Int,(Int,Int))

kruskal :: [City] -> [(City,City)] -> IO Int
kruskal nodes edges = do
  let sorted = sortBy comp (map (\((i,ci),(j,cj)) -> (i,j,dist ci cj)) edges)

  uf <- newUFIO (length nodes)

  let
    kruskal' e@(i,j,c) = do
      same <- sameUFIO i j uf
      if same then
        return 0
        else do
        uniteUFIO i j uf
        return c
        
  mapM kruskal' sorted >>= return . sum
  where
    comp (_,_,c1) (_,_,c2) = compare c1 c2
    dist (x1,y1) (x2,y2) = min (abs $ x1-x2) (abs $ y1-y2)

listToTuple2 (a:b:_) = (a,b)

main = do
  n <- readLn :: IO Int
  nodes <- BS.getContents >>= return . zip [1..] . map (listToTuple2 . map (fst . fromJust . BS.readInt) . BS.words) . BS.lines :: IO [(Int,(Int,Int))]

  
  let roads = f (sortBy (comparing (fst . snd)) nodes) ++ f (sortBy (comparing (snd . snd)) nodes)

  print =<< kruskal nodes roads

  where
    f (_:[]) = []
    f (x:y:xs) = (x,y) : f (y:xs)